

JUEGOS DE ORDENADOR

# ESPIAS

..COMMODORE 64...VIC 20...APPLE...SPECTRUM...DRAGÓN...MSX...



EDICIONES GENERALES ANAYA

Grupo distr. edit. 212.  
F. 11.86

# ESPIAS

Jenny Tyler y Chris Oxlade

Ediciones Generales Anaya



## Contenido

- 2 Acerca de este libro
- 3 Ojos de espía
- 4 El faro rastreador
- 6 El robot-espía
- 8 Examen de espía
- 10 El codificador  
de mensajes secretos
- 12 La cita
- 16 El entrenador Morse
- 17 Respuestas a las  
cuestiones planteadas

# Acerca de este libro

Los programas de este libro están escritos en una versión estándar de BASIC, uno de los lenguajes de ordenador, y hay líneas de conversión para la mayoría de los principales tipos de ordenadores personales. Busca a la izquierda del listado del programa el símbolo que representa a tu ordenador y después mira en la lista de cambios los necesarios para la versión correcta de esa línea. Los símbolos para los distintos ordenadores son los siguientes:

- ▲ Commodore 64 y VIC 20
- Spectrum
- Apple
- ★ DRAGON
- MSX

## Acerca de los juegos

Los juegos de este libro son muy sencillos. Se han creado con la intención de ayudarte a acostumbrarte a tu ordenador y al lenguaje BASIC, tecleando los programas, depurándolos y comprendiendo cómo funcionan. Los programas no tienen gráficos o sonido porque varían mucho de un ordenador a otro, pero puedes intentar añadirselos.

Puedes cambiar y adaptar los juegos a tu gusto. Detrás de cada listado de programa se sugieren maneras de hacerlo, pero también puedes experimentar con tus propias ideas. De esta manera, puedes usar los juegos de este libro como base para conseguir juegos más largos y más complicados hechos por ti mismo.

## Teclear y ejecutar los programas

Recuerda que puede resultar difícil y llevar mucho tiempo teclear correctamente los programas, incluso si éstos son cortos. Comprueba cada línea que introduzcas, ya que es fácil cometer errores aunque tengas bastante experiencia.

Cuando hayas introducido la totalidad del programa, compruébalo otra vez, asegurándote de que no te has saltado ninguna línea, espacios o signos de puntuación.

Para empezar el juego, teclea RUN. Antes de empezar, lee la introducción del juego para tener una idea de lo que se supone que éste debe hacer. Si el programa no funciona correctamente, es bastante probable que haya un error en cualquier sitio; por eso, lista el programa y compruébalo de nuevo.

Al acabar el juego, puede que el ordenador te pregunte si quieres jugar otra vez o, por el contrario, decir algo como BREAK in 200, en cuyo caso debes teclear RUN para jugar otra vez.

## Cambiar la velocidad

Algunos juegos dependen de la velocidad de tus reacciones y del ordenador. Puede ser necesario ajustar la velocidad. Encontrarás instrucciones para hacerlo a continuación del listado del programa.

*El robot espía fue escrito por Adrian Hall.*

*Ilustrado por Sue Walliker, Rob McCaig y Martin Newton.  
Diseñado por Roger Priddy.*

*Queda prohibida la reproducción total o parcial de la presente obra bajo cualquiera de sus formas, gráficas o audiovisuales sin la autorización previa y escrita del editor, excepto citas en revistas, diarios o libros, siempre que se mencione la procedencia de las mismas.*

Título original: Computer Spy Games

© 1984 Usborne Publishing Ltd.

© 1985, de la edición española.

Ediciones Generales Anaya

Villafranca, 22. 28028 Madrid

ISBN: 84-7525-309-1


Depósito legal: M. 39.276/1985

Impreso por Melsa

Carretera de Fuenlabrada a Pinto, km. 21.800

Pinto (Madrid)

Impreso en España - Printed in Spain



# Ojos de espía

Si crees que eres un buen espía, prueba este juego.

El ordenador imprimirá los números 1 a 9 en tu pantalla. Obsérvalos como un halcón mientras pulsas una tecla (vale cualquiera). Uno de ellos se mueve, pero ¿cuál? Cuando creas que lo sabes, pulsa otra vez una tecla y díselo al ordenador.

Comprueba que tu capacidad de observación no es tan buena como pensabas.

```

10 DIM X(9):DIM Y(9)
20 LET P=0
30 FOR I=1 TO 9
40 GOSUB 340:LET X(I)=M+3
50 GOSUB 340:LET Y(I)=M+3
60 NEXT I
70 GOSUB 360
80 GOSUB 310
90 GOSUB 340
100 LET M=M:GOSUB 340
110 LET X(M)=X(M)+SIGN(M-5.1)
120 GOSUB 360
130 GOSUB 310
●▲ 140 CLS:PRINT
150 PRINT "QUE NUMERO SE HA MOVIDO ?"
160 INPUT A
170 IF A(>)M THEN GOTO 250
●▲ 180 CLS:PRINT
190 PRINT "BIEN ESPIAO!"
200 LET P=P+1
210 PRINT "AHORA TIENES '¡P!' PUNTOS"
220 PRINT :PRINT "PULSA UNA TECLA"
230 GOSUB 310
240 GOTO 30
●▲ 250 CLS:PRINT :PRINT "EQUIVOCADO-FIN DEL JUEGO"
260 PRINT "LA RESPUESTA CORRECTA ES '¡M'"
270 PRINT "TU TANTEO '¡P!' PUNTOS"
280 PRINT "OTRA PARTIDA ? (S/M)"
290 INPUT A$:IF A$="S" THEN RUN
300 STOP
●▲ 310 LET I$=INKEY$
320 IF I$="" THEN GOTO 310
330 RETURN
□■★ 340 LET N=INT(RND(1)*9)+1
350 RETURN
●▲ 360 CLS
370 FOR I=1 TO 9
□■▲●★ 380 PRINT TAB(X(I),Y(I));STR$(I)
390 NEXT I
400 RETURN

```

## Cómo funciona el programa

10: Define el espacio para las variables que contienen las posiciones de los números.

20: Inicializa la variable del tanteo a cero.

30-60: Elige las posiciones de los números.

70: Salta a la subrutina de impresión de los números.

80: Salta a la subrutina que espera la pulsación de una tecla.

90-100: Salta a la subrutina que elige uno de los números y la llama M.

110: Elige el movimiento a derecha o izquierda.

120: Salta a la subrutina que imprime los números.

130: Salta a la subrutina que espera la pulsación de una tecla.

140-150: Borra la pantalla. Imprime un mensaje.

160: Coge tu número.

170: Si te has equivocado, el programa salta a la línea 250 para darte el mensaje.

180-190: Borra la pantalla e imprime un mensaje.

200-210: Incrementa el tanteo y lo imprime.

220-230: Imprime un mensaje y salta a la subrutina que espera la pulsación de una tecla.

240: Vuelve para la siguiente baza.

250-300: Si te has equivocado, imprime el mensaje y te pregunta si quieres jugar otra vez.

310-330: Subrutina que espera la pulsación de una tecla.

340-350: Subrutina para elegir un número aleatorio entre 1 y 9.

360-400: Subrutina que imprime los números en la pantalla en las posiciones correspondientes.


## Líneas de conversión

- 140,180,250,360 Sustituir CLS por HOME
- ▲ 140,180,250,360 Sustituir CLS por PRINT CHR\$(147)
- 310 I\$="" : IF PEEK(-16384)>127 THEN GET I\$
- ▲ 310 GET I\$
- 340 Sustituir RND(1) por RND
- ★ 340 Sustituir RND(1) por RND(-TIME)
- 340 LET N=RND(9)
- 380 PRINT AT Y(I),X(I);STR\$(I)
- ▲ 380 PRINT CHR\$(19);FOR LL=1 TO Y(I):PRINT:NEXT:PRINT TAB(X(I));STR\$(I).
- 380 PRINT Y(I)+32+X(I),STR\$(I);
- ★ 380 LOCATE X(I),Y(I):PRINT STR\$(I)



Intenta hacer una subrutina que conserve la puntuación más alta obtenida hasta el momento.

# El faro rastreador



Esta misma noche debe llevarse a cabo una misión absolutamente secreta y desesperadamente peligrosa. ¡Y debes realizarla tú!

Debes cruzar una sección estrechamente vigilada del territorio enemigo y regresar, evitando su potente y gigantesco faro rastreador. Hay rocas, arbustos y otros obstáculos donde esconderse, pero no hay segundas oportunidades: una vez te hayan visto, habrás perdido la que tenías. Cuando hayas completado una misión con éxito, hay otra, más difícil incluso, que llevar a cabo. Poder seguir sólo depende de tu habilidad.

Usa la tecla M para moverte a la derecha y la N para moverte a la izquierda. Para completar una misión, debes cruzar la pantalla de izquierda a derecha y regresar de nuevo.

```
10 GOSUB 450
```

```
20 LET A=1:LET B=0:LET S=0
```

```
●▲ 30 CLS
```

```
40 LET X=0:LET Y=12:LET B=A*(A)
```

```
50 GOSUB 380
```

```
60 LET F=0:LET N=0:LET NN=0:GOSUB 340
```

```
70 LET L=0:LET C=0:LET TC=10:LET CI=0
```

```
●▲ 80 LET I$=INKEY$
```

```
90 IF I$="N" THEN LET NN=N-1
```

```
100 IF I$="M" THEN LET NN=N+1
```

```
110 IF NN>19 THEN LET NN=19
```

```
120 IF NN<0 THEN LET NN=0
```

```
130 IF NN=19 AND F=0 THEN LET F=1
```

```
140 IF NN=0 AND F=1 THEN LET F=2
```

```
150 GOSUB 340
```

```
160 IF N<>NN THEN LET S=S+1
```

```
170 LET N=NN:LET B=B+1
```

```
180 GOSUB 400
```

```
□ 190 IF MID$(A$(A),N+1,1)="" AND L=1 THEN GOTO 240
```

```
200 FOR T=1 TO 50:NEXT T
```

```
210 IF F<2 THEN GOTO 80
```

## Cómo funciona el programa

10: Salta a la subrutina que lee los datos.

20: Inicializa variables.

40-50: Imprime los obstáculos.

60-70: N es la posición del jugador. NN es la nueva posición.

80: Comprueba si has pulsado alguna tecla.

90: Si la tecla pulsada es N (izquierda), la nueva posición es N menos 1. Si es M (derecha), es N más 1.

100-120: Interrumpe el movimiento del jugador cuando llega al final.

130-140: F es 1 cuando el jugador llega al extremo derecho de la pantalla, y 2 cuando vuelve a alcanzar la parte izquierda.

150: Salta a la subrutina que mueve al jugador.

160: Si el jugador se ha movido, se incrementa S.

170: Hace la posición del jugador igual a la nueva posición e incrementa G. G actúa como contador; se incrementa siempre, tanto si se mueve el jugador o no.

180: Salta a la subrutina que enciende y apaga la luz.

190: Comprueba si el jugador está debajo de un espacio cuando se enciende la luz y si es así, salta para imprimir el mensaje de que ha perdido.

200: Pausa.

210: Si el jugador no ha regresado a la parte izquierda, regresa para comprobar si pulsa alguna tecla.

¿Puedes hacer que la luz permanezca más tiempo encendida?

Quizá puedas sustituir los obstáculos, que vienen como «=» en el listado, por símbolos gráficos para representar rocas, árboles, edificios o cualquier otra cosa que te guste.



```

220 LET A=A+1:IF A=8 THEN LET A=7
230 GOTO 30
240 LET X=4:LET Y=1:LET B$='ACADAS DE SER DESCUBIERTO'
250 GOSUB 380:PRINT
260 PRINT "TU TANTEO ";INT((A-1+S/G)*100)
270 PRINT:PRINT "OTRA VEZ? (S/N)"
280 INPUT C$:IF C$='S' THEN RUN
290 PRINT "ADIOS.....":STOP
300 LET Y=3:LET X=10:LET B$='?'
310 GOSUB 380:RETURN
320 LET X=10:LET Y=3:LET B$=' '
330 GOSUB 380:RETURN
340 LET X=N:LET Y=13:LET B$=' '
350 GOSUB 380
360 LET X=NN:LET B$='S'
370 GOSUB 380:RETURN
★□■●▲ 380 PRINT TAB(X,Y);B$
390 RETURN
400 IF L=1 THEN LET C=C+1
★□■ 410 IF C=TC THEN LET L=0:LET C=0:LET TC=INT(RND(1)*8+(12-A)):GOSUB 320
420 IF L=0 THEN LET C1=C1+1
★□■ 430 IF C1=TC THEN LET L=1:LET C1=0:LET TC=INT(RND(1)*10+(8-A)):GOSUB 300
440 RETURN
□ 450 DIM A$(7)
460 FOR I=1 TO 7: READ A$(I):NEXT I
470 RETURN
480 DATA "== == == == == == =="
490 DATA "== == == == == == =="
500 DATA "= == == == == == =="
510 DATA "= == == == == == =="
520 DATA "= == == == == == =="
530 DATA "= == == == == == =="
540 DATA "= == == == == == =="

```



Añade sonidos de alarma para cuando el enemigo descubre al jugador.

220: Incrementa el nivel de dificultad.  
230: Vuelve a imprimir obstáculos nuevos.  
240-290: Imprime el mensaje que te indica que has perdido, así como el tanteo, y pregunta si quieres jugar otra vez.  
300-310: Imprime el foco de búsqueda.  
320-330: Apaga el foco.  
340-370: Subrutina que actualiza la posición del jugador.  
380-390: Subrutina que imprime la variable B\$ en la pantalla, en la posición X,Y.  
400-440: Subrutina que determina el tiempo que está encendida y apagada la luz.  
450-470: Subrutina que lee los datos de los obstáculos.  
480-540: Líneas de datos.

Puedes hacer el juego más fácil dando al jugador una segunda oportunidad. ¿Sabes cómo hacerlo?



### Líneas de conversión

- 30 HOME
  - ▲ 30 PRINT CHR\$(147)
  - ▲ 80 GET I\$
  - 80 I\$="":IF PEEK(-16384)>127 THEN GET I\$
  - 190 IF A\$(A,N+1)=" " AND L=1 THEN GOTO 240
  - 380 PRINT AT Y,X;B\$
  - 380 VTAB(Y):HTAB(X+1):PRINT B\$
  - ▲ 380 PRINT CHR\$(19):FOR LL=1 TO Y:PRINT:NEXT:PRINT TAB(X);B\$
  - 380 PRINT# Y#32+X,B\$;
  - ★ 380 LOCATE X,Y:PRINT B\$
  - 410,430 Sustituir RND(1) por RND
  - 410,430 Sustituir RND(1) por RND(0)
  - ★ 410,430 Sustituir RND(1) por RND(-TIME)
  - 450 DIM A\$(7,20)
- La línea 200 controla la velocidad

Puedes cambiar las posiciones de los obstáculos cambiando las líneas de datos.



# El robot-espía



Estás a cargo del control del robot-espía, un aparato único que funciona por control remoto y que sigue en secreto a los agentes enemigos. Tú recibes detalles de los movimientos de un agente por las calles —si gira a la derecha o a la izquierda— y debes repetir estos movimientos con exactitud cuando envíes señales al robot-espía, de tal manera que pueda permanecer en contacto con el agente.

Desafortunadamente, el agente sabe que el robot-espía lo está siguiendo. Por eso, hace que tu trabajo sea tanto más complicado cuanto más tiempo lo estés siguiendo. También ha saboteado tu aparato de control remoto, reorganizando las teclas en un intento de confundirte. Esto hace que tú tengas que pulsar I (de izquierda) con tu mano derecha y D (de derecha) con la izquierda. ¿Podrás con él, o será él quien gane la partida?

```

10 LET SP=.3
●▲ 20 CLS
30 LET H=0
40 LET K=0:LET S=0:LET L=1
50 PRINT ">>>ROBOT ESPIA<<<":PRINT
60 PRINT "PULSA E PARA EMPEZAR"
●▲ 70 LET C$=INKEY$
80 IF C$("<>E") THEN GOTO 70
●▲ 90 CLS:PRINT:PRINT
100 IF K=5 THEN LET L=L+1:LET K=0
110 LET M$=""
120 LET C$=""
130 FOR T=1 TO L
□■★ 140 LET N=RND(1)
150 IF N>=.5 THEN PRINT "IZQUIERDA":LET M$=M$+"I"
160 IF N<.5 THEN PRINT "DERECHA":LET M$=M$+"D"
170 NEXT T
180 LET K=K+1
190 FOR D=1 TO 200:L*SP
●▲ 200 LET C$=INKEY$
210 NEXT D
●▲ 220 CLS:PRINT:PRINT
230 LET E=0
240 PRINT "Que direccion/es era/n?"
250 FOR T=1 TO L
260 PRINT "?";
●▲ 270 LET C$=INKEY$
280 IF C$("<>D") AND C$("<>I") THEN GOTO 270
290 PRINT C$
□ 300 IF C$("<>MID$(M$,T,1) THEN LET E=1
310 NEXT T
320 IF E=0 THEN LET S=S+L
330 IF S>=H THEN LET H=S
340 IF E=1 THEN PRINT "LE PERDISTE! VAYA ESPIA!"
350 PRINT "PUNTUACION = ";S
360 PRINT "RECORD DE PUNTUACION = ";H
370 IF E=1 THEN GOTO 40
380 FOR D=1 TO 500:L*SP
390 NEXT D
400 GOTO 90

```

¿Puedes hacer que el ordenador pregunte el nombre al jugador cuando se alcanza un nuevo mejor tanteo?



## Cómo funciona

30-40: Inicializa variables. L es el número de palabras impresas en la pantalla. K cuenta cuántas veces se han impreso L palabras.

60-80: Imprime un mensaje y espera a que se pulse una tecla para empezar.

90: Borra la pantalla y deja dos líneas vacías.

100: Si K es 5, se incrementa el valor de L y K se hace cero.

110-120: Define dos variables de cadena vacías.

130-170: Bucle de L repeticiones eligiendo la secuencia de derechas e izquierdas aleatoriamente, imprimiéndolas y guardándolas en M\$.

180: Incrementa K.

190-210: Pausa para que el jugador vea las palabras. Es más larga cuantas más palabras haya.

230: Variable de error: será 1 si el jugador ha tenido un fallo.

250: Comienza el bucle de captura y comprobación de las respuestas.

260-270: Imprime el *prompt*, comprueba si se pulsa una tecla y la pone en C\$.

280: Si la tecla pulsada no es I o D vuelve por otra.

300: Comprueba si la letra de C\$ coincide con la correspondiente de M\$ y si no, pone a uno la variable de error.

320: Actualiza el tanteo si no hay ningún fallo.

330: Si es necesario, actualiza el mejor tanteo.

340: Si hay algún fallo, imprime el mensaje.

350-360: Imprime el tanteo.

370: Vuelve para jugar otra vez.

380-390: Pausa para ver el tanteo.

400: Regresa para la siguiente baza.

## Líneas de conversión

- 20,90,220 Sustituir CLS por HOME
- ▲ 20,90,220 Sustituir CLS por PRINT CHR\$(147)
- 70,200,270 C\$="":IF PEEK(-16384)>127 THEN GET C\$
- ▲ 70,200,270 GET C\$
- 140 LET N=RND
- 140 LET N=RND(0)
- ★ 140 LET N=RND(-TIME)
- 300 IF C\$("<>M\$(T) THEN LET E=1

La línea 10 controla la velocidad del juego, cambia este número de la manera siguiente:  
Spectrum 0.2, VIC-20 0.4,  
MSX 0.5, Apple 0.1.  
Cualquiera que sea el ordenador que uses, cuanto más bajo sea el número más rápido será el juego. Prueba a ir aumentando la velocidad según juegas para ver cómo eres de bueno.



# Examen de espía

Como recluta en la Escuela de Espías, tienes que pasar un montón de exámenes Q para graduarte. Empiezas en el nivel más bajo; el Grado 5 de Aprendiz de Espía. Tu meta es alcanzar lo más alto y llegar a ser un VIS (Very Important Spy = Espía Muy Importante) de Grado 1 e incluso lograr el máximo: el Diploma de Super Espía.

En cada examen Q aparecen 10 posiciones en la pantalla de tu ordenador y se te dan números entre 1 y 100. Tu meta es colocar en orden estos números en las 10 posiciones, estando el más bajo en la posición 1 y el más alto en la 10.

Puedes descartar algunos números si no tienes dónde colocarlos, pulsando la tecla D. El número de números que puedes desechar es el mismo que el número de tu grado.

## Cómo funciona el programa

```
□ 10 DIM N(10):DIM N$(5)
●▲ 20 CLS
30 GOSUB 510,
40 LET W$=""
50 LET D=5
60 LET G=0
70 FOR I=1 TO 10:LET N(I)=0:NEXT I
80 LET I=1
90 GOSUB 430
★■□ 100 LET M=INT(RND(1)*99+1)
110 PRINT:PRINT "DONDE PONES ";M
120 PRINT:INPUT P$
130 IF P$="D" AND G<D THEN LET G=G+1:GOTO 90
140 IF P$="D" THEN PRINT "NO PUEDES!":GOTO 120
150 LET P=VAL(P$)
160 IF P<1 OR P>10 THEN GOTO 120
```

10: Dimensiona las tablas.

20: Borra la pantalla.

30: Salta a la subrutina que lee los datos y los pone en la tabla N\$().

40: Inicializa WS.

50: Hace 5 el Grado al comienzo del programa.

60: G es el número de números descartados.

70: Pone ceros en las 10 posiciones para empezar.

80: Cuenta los números que el jugador ha colocado.

90: Salta a la subrutina que imprime tu nivel, los números 1 a 10 y los que ya hubiera en esas posiciones.

100: Elige un número entre 1 y 100.

110-120: Imprime el número y averigua dónde quieres ponerlo.

130-140: Comprueba si pulsas D y si puedes o no descartar algún número.

150: Calcula el valor numérico de tu entrada de datos.

160: Vuelve a pedirte el dato si el número no está entre 1 y 10.

¿Puedes hacer que el número de posiciones sea 15 en vez de 10?



Cambia los nombres si quieres.

¿Serás capaz de llegar a Superespía?



```

170 IF N(P)>0 THEN PRINT "YA ESTA LLENO":GOTO 120
180 LET N(P)=M
190 LET F=0
200 FOR L=P TO 10
210 IF N(L)<M AND N(L)<>0 THEN LET F=1
220 NEXT L
230 FOR L=1 TO P
240 IF N(L)>M AND N(L)<>0 THEN LET F=1
250 NEXT L
260 IF F=1 THEN GOTO 360
270 LET I=I+1:IF I<11 THEN GOTO 90
280 LET D=D-1:IF D=0 THEN GOTO 330
290 PRINT "BIEN HECHO, PASAS AL GRADO ";D
300 PRINT:PRINT "AHORA ERES UN ";N$(D)
310 LET W$=""
320 GOTO 400
330 PRINT "ASOMBROSO! HAS ALCANZADO"
340 PRINT "EL GRADO DE SUPER ESPIA"
350 STOP
360 PRINT "MAL! NO ERES BASTANTE BUENO"
370 PRINT:PRINT "AUN ERES UN ";
380 PRINT N$(D)
390 LET W$="AUN"
400 PRINT:PRINT "QUIERES INTENTARLO OTRA VEZ? (S/N)"
410 INPUT A$:IF A$="S" THEN GOTO 60
420 STOP
●▲ 430 CLS
440 PRINT:PRINT"ERES ";W$;" UN ";N$(D)
450 PRINT
460 FOR J=1 TO 10
470 PRINT J;
480 IF N(J)>0 THEN PRINT N(J);
490 PRINT:NEXT J
500 RETURN
510 FOR I=1 TO 5:READ N$(I)
520 NEXT I
530 RETURN
540 DATA "VIS","ESPIA","ESPIA JUNIOR"
550 DATA "ASISTENTE DE ESPIA","APRENDIZ DE ESPIA"

```

170: Hace lo mismo si la posición está ya ocupada.

180: Pone el número en la posición que tú has indicado.

190-220: Comprueba todas las posiciones siguientes a la que has dicho para ver si hay ya un número más bajo en alguna de ellas. Si es así hace F=1.

230-250: Comprueba las posiciones anteriores y pone F a 1 si hay ya un número más alto en alguna de ellas.

260: Si la posición no es correcta, salta a otra línea para indicártelo.

270: Incrementa el contador y regresa a por otro número.

280-300: Cambia D al Grado siguiente e imprime el mensaje. (Va a Superespía si D es 0.)

310: Hace W\$ igual a una cadena vacía.

320: Salta para imprimir el mensaje «inténtalo otra vez».

330-350: Mensaje de Superespía.

360-380: Imprime el mensaje de que has suspendido.

390: Hace W\$ igual a «todavía».

400-420: Averigua si quieres intentarlo otra vez; si no, interrumpe el juego.

430-500: Subrutina que imprime mensajes y números.

510-530: Subrutina que lee los nombres de los Grados y los pone en la tabla NS.

540-550: Datos.


### Líneas de conversión

- 10 DIM N(10):DIM N\$(5,16)
- 20,430 Sustituir CLS por HOME
- ▲ 20,430 Sustituir CLS por PRINT CHR\$(147)
- 100 LET N=INT(RND\*99+1)
- 100 LET N=RND(99)
- ★ 100 LET N=INT(RND(-TIME)\*99+1)



# El codificador de mensajes secretos

Usa este programa para enviar mensajes codificados a tus amigos. Ellos necesitarán un ordenador para decodificarlos, aunque no necesariamente del mismo tipo que el tuyo. Pero no olvides que necesitarán una copia del programa. Así, podrán decodificar tus mensajes y también enviártelos codificados.



Puedes establecer una red de espionaje con todos tus amigos que tengan ordenador.

## Cómo funciona el programa

10-80: Imprime el «menú» y pregunta si quieres codificar o decodificar.

90-100: Salta a la subrutina correspondiente.

110: Regresa otra vez al menú.

120-200: SUBROUTINA DE CODIFICACIÓN.

120: Salta a la subrutina que pide el mensaje.

130: Elige una cantidad para cambiar las letras y la pone en X.

140: Añade una letra en caso de que la primera palabra tenga un solo carácter.

150: Salta a la subrutina para cambiar X letras del alfabeto y colocarlas al revés.

160: Añade una letra al mensaje para decir a la subrutina decodificadora cuál es el valor de X.

170: Comprueba si el mensaje tiene un número par de caracteres. Si es así, salta a la subrutina que intercambia las letras de cada par.

210-300: SUBROUTINA DE DECODIFICACIÓN.

210: Salta a la subrutina que pide el mensaje.

220: Si el mensaje contiene un número par de letras, las intercambia en cada par.

230-240: Extrae la primera letra. (Es la añadida en la línea 160, que indica la cantidad de cambios.)

250: Calcula el valor de X.

260: Modifica la dirección del cambio y salta a la subrutina que invierte las letras.

```
●▲ 10 CLS:PRINT:PRINT "CODIFICADOR DE MENSAJES SECRETOS"
```

```
20 PRINT"====="
```

```
30 PRINT:PRINT "QUE QUIERES : "
```

```
40 PRINT
```

```
50 PRINT " 1.CODIFICAR UN MENSAJE"
```

```
60 PRINT " 0 2.DECODIFICAR UN MENSAJE"
```

```
70 PRINT:PRINT
```

```
80 PRINT "INTRODUCE UN NUMERO":INPUT A
```

```
90 IF A=1 THEN GOSUB 120
```

```
100 IF A=2 THEN GOSUB 210
```

```
110 GOTO 30
```

```
120 LET C$="CODIFICAR":GOSUB 400
```

```
★■□ 130 LET X=INT(RND(1)*25+1)
```

```
140 LET M$="F"+M$
```

```
150 GOSUB 310:GOSUB 420
```

```
160 LET M$=CHR$(X+64)+M$
```

```
170 IF LEN(M$)/2=INT(LEN(M$)/2) THEN GOSUB 450
```

```
180 PRINT "EL MENSAJE CODIFICADO ES: "
```

```
190 PRINT M$
```

```
200 RETURN
```

```
210 LET C$="DECODIFICAR":GOSUB 400
```

```
220 IF LEN(M$)/2=INT(LEN(M$)/2) THEN GOSUB 450
```

```
□ 230 LET K$=LEFT$(M$,1)
```

```
□ 240 LET M$=RIGHT$(M$,LEN(M$)-1)
```

```
□ 250 LET X=ASC(K$)-64
```

```
260 LET X=-X:GOSUB 420
```

XOFJ PDG ETHEDGGG  
TDH ARPJHUQ

2 espacios  
CVLURE QH UDLIQRF  
HEHG HV RQI

CTFF UXYJ BC WDICY  
PYFKOZU

EMMMMLLLLFF XJ TUTY  
QJK

SHWGTKBF TMLX XM  
GXBNZETY

2 espacios  
BOGDH TTCJXUF



- 270 GOSUB 310:M\$=RIGHT\$(M\$,LEN(M\$)-1)
- 280 PRINT "EL MENSAJE DECODIFICADO ES: "
- 290 PRINT M\$
- 300 RETURN
- 310 LET N\$=""
- 320 FOR I=1 TO LEN(M\$)
- 330 LET Q\$=MID\$(M\$,I,1):LET N=ASC(Q\$)
- 340 IF N=32 THEN GOTO 380
- 350 LET N=N+X
- 360 IF N>90 THEN LET N=N-26
- 370 IF N<65 THEN LET N=N+26
- 380 LET M\$=M\$+CHR\$(N):NEXT I
- 390 LET M\$=M\$:RETURN
- 400 PRINT "CUAL ES EL MENSAJE PARA "IC\$
- 410 INPUT M\$:RETURN
- 420 LET N\$="":FOR I=LEN(M\$) TO 1 STEP -1
- 430 LET N\$=N\$+MID\$(M\$,I,1):NEXT I
- 440 LET M\$=N\$:RETURN
- 450 LET N\$="":LET N=LEN(M\$)
- 460 FOR I=1 TO LEN(M\$)-1 STEP 2
- 470 LET N\$=N\$+MID\$(M\$,I+1,1)
- 480 LET N\$=N\$+MID\$(M\$,I,1)
- 490 NEXT I:LET M\$=N\$:RETURN

270: Salta a la subrutina que cambia las letras en el alfabeto. Separa la primera letra.

310-390: SUBROUTINA DE CAMBIO.

310: Pone el mensaje en NS.

320: Comienzo del bucle, que toma una letra de NS cada vez.

330: Calcula el número de código de la letra para el ordenador.

340: Si hay un espacio, lo mantiene igual.

350: Añade el valor de X al código de la letra.

360-370: Ajusta el número de código si va más allá del código de la Z o es menor que el de la A.

380: Añade una nueva letra a NS y vuelve al principio del bucle.

390: Cuando se han cambiado todas las letras, se pone el nuevo mensaje en MS.

400-410: Subrutina que pide el mensaje.

420-440: Subrutina que invierte las letras.


450-490: Subrutina que toma por orden pares de letras del mensaje y las intercambia dentro de la misma pareja.

Los buenos espías normalmente comprueban que sus mensajes pueden decodificarse correctamente antes de enviarlos.



### Líneas de conversión

- 10 Sustituir CLS por HOME
- ▲ 10 Sustituir CLS por PRINT CHR\$(147)
- 130 LET X=INT(RND\*25)+1
- 130 LET X=RND(25)
- ★ 130 LET X=INT(RND(-TINE)\*25)+1
- 230 LET K=M\$(1)
- 240 LET M\$=M\$(2 TO )
- 250 LET X=CODE(K\$)-64
- 270 GOSUB 310:LET M\$=M\$(2 TO )
- 330 LET Q\$=M\$(I):LET N=CODE(Q\$)
- 430 LET N\$=N\$+M\$(I):NEXT I
- 470 LET N\$=N\$+M\$(I+1)
- 480 LET N\$=N\$+M\$(I)



XLYRLCGK ML <sup>!espacio</sup>CSO QMPRM  
QMJ Y GBD

PIEDQBF JEB UDUYJ  
BHQSV

JKCKM SW K OD VO  
BKWYD K XOFF

QLOTZAM EF QAEQYQO  
HAERZ

¿Puedes averiguar lo que están diciendo?



# La cita

Tu misión es complicada; por eso, lee estas instrucciones cuidadosamente.

Debes recoger una maleta en una caja de seguridad de la estación, entregársela a tu contacto y regresar al aeropuerto antes de que despegue el último avión (tu ordenador te dirá la hora).


También te dirá dónde encontrar a tu contacto y a qué hora. Por eso, debes dejar un mensaje en ese lugar antes de que él llegue, diciéndole dónde y cuándo te verás con él para darle la maleta.

Debes averiguar la palabra clave antes de verlo, y asegúrate de que no llegas más de 15 minutos tarde.

Antes de que puedas coger la maleta, debes encontrar la llave de la caja y también su número. Por desgracia, la llave está en manos de agentes enemigos, cuyo cuartel general está en el hotel. Debes encontrar un espía enemigo y seguirlo, esperando que no sea muy cuidadoso y pierda la llave (y, desde luego, que no te vea).

El plano te muestra los sitios a los que puedes ir y la lista siguiente, las palabras que puedes usar en el juego.

Adónde puedes ir:  
AEROPUERTO



Ahora cómete el libro.

Puede que necesites tomar nota de la palabra clave y del número de la caja (codificado por supuesto).

## Palabras que puedes usar:

HORA—Te dice qué hora es.

IR—Te pregunta adónde. Puedes ir a cualquier lugar indicado en el plano.

DECIR—La palabra clave.

EXAMINAR—Cualquier cosa. (Examina la llave para saber el número.)

LEER—Un mensaje.

ABRIR—La caja de seguridad.

SEGUIR—A un espía enemigo.

ESPERAR—Durante un cierto tiempo.

DEJAR—Un mensaje.


BUSCAR—En cualquier sitio (para encontrar la llave).

AYUDA—Te recuerda la hora y el lugar de encuentro.

También puedes usar cualquiera de los nombres del plano que están en letras mayúsculas.

Si quieres, cambia las palabras clave por otras cualesquiera.

También puedes cambiar los nombres de los lugares.



Si usas un VIC-20, necesitarás la memoria extra para este juego.

```

■ □ 10 DEF FMA(X)=INT(RND(1)*X)+1
    20 GOSUB 1010:GOSUB 1130
    30 LET P=1:LET Y=0:LET X=1:LET EP=10
    40 LET U=0:LET NM=0
    50 LET S$=""
    60 LET B$=""
● ▲ 70 CLS:PRINT:PRINT 'CITA'
    80 PRINT '===='
    90 LET NM=NM+1
    100 PRINT:PRINT
    110 PRINT B$:PRINT
    120 PRINT:PRINT 'ESTAS EN '
    130 PRINT R$(P)
    140 IF EP=P THEN PRINT 'EL AGENTE ENEMIGO ESTA AQUI':LET NE=NE+1
    150 IF EP(>)P THEN LET NE=0
    160 IF P=HP AND F(1)=0 THEN PRINT 'MENSAJE PARA TI AQUI'
    170 LET F(3)=0
    180 LET T1=H*M/100

```

## Cómo funciona el programa

10: Define una función que elige un número aleatorio entre 1 y X.

20: Salta al final del programa para leer los datos e inicializar el juego.

30: P, Y y X registran la posición del jugador. EP es la posición del enemigo.

60: BS sirve para los mensajes que el ordenador da al jugador.

90: Incrementa NM (número de movimientos).

140: NE cuenta cuánto tiempo está el jugador en el mismo lugar que el agente enemigo.

180: Calcula la hora (H= horas, M= minutos).

```

190 IF F(4)=1 AND R$(P)=S$ AND U<=T1 AND U+.15>T1 THEN PRINT T$:LET F(3)=1
200 IF P=1 AND H<FH AND F(7)=1 THEN GOTO 890
210 PRINT:PRINT:PRINT "QUE ES LO SIGUIENTE"
220 LET B$=""
230 INPUT I$
240 LET V=0:FOR I=1 TO 11
250 IF I$=V$(I) THEN LET V=I
260 NEXT I
270 IF V=0 THEN LET V=12
280 IF NE=3 AND FNA(10)>3 THEN LET B$="EL AGENTE ENEMIGO TE ESTA VIENDO !":GOTO 70
290 IF NE=4 AND V<>I THEN PRINT "TE HAN CAPTURADO !":STOP
300 ON V GOSUB 360,420,490,540,570,640,710,730,780,810,820,870
310 LET M=M+DT:IF M>59 THEN LET M=M-60:LET H=M+1
320 IF F(2)=1 AND H)=CH THEN LET F(4)=1
330 IF H=FH THEN GOTO 880
340 IF FNA(10)>9 THEN LET EP=10
350 GOTO 70
360 PRINT:PRINT "A DONDE":INPUT M$
370 LET NP=0:FOR I=1 TO 20
380 IF M$=R$(I) THEN LET NP=I
390 NEXT I:IF NP=0 THEN GOTO 360
400 GOSUB 950
410 LET P=NP:RETURN
420 LET DT=5
430 PRINT:PRINT "DECIR BUE":INPUT Q$
440 IF EP=P THEN LET B$="HAS ALCANZADO AL AGENTE ENMIGO !":RETURN
450 IF F(3)=0 THEN LET B$="NADIE TE OYE":RETURN
460 IF Q$<>P$ THEN LET B$="EL CONTACTO TE IGNORA":RETURN
470 IF F(6)=1 THEN LET B$="HICISTE EL CONTACTO-TIENE LA MALETA !"
480 LET F(7)=1:RETURN
490 LET DT=5
500 PRINT:PRINT "QUE QUIERES EXAMINAR":INPUT Q$
510 IF Q$="MALETA" THEN LET B$="TOP SECRET !":RETURN
520 IF Q$="LLAVE" THEN LET B$="UN NUMERO - "+STR$(NL):RETURN
530 LET B$="NADA ESPECIAL !":RETURN
540 IF P<>NP OR F(1)=1 THEN LET B$="NADA PARA LEER !":RETURN
550 LET B$="UNA PALABRA - '"+P$+'"'
560 LET F(1)=1:RETURN
570 LET DT=5
580 IF P<>16 THEN LET B$="NADA PARA ABRIR":RETURN
590 IF F(5)=0 THEN LET B$="NO TIENES LLAVE":RETURN
600 PRINT:PRINT "QUE NUMERO":INPUT YN
610 IF NL<>YN THEN LET B$="NO ES ESA LLAVE":RETURN
620 LET B$="LA CAJA ESTA ABIERTA-TIENES LA MALETA":LET F(6)=1
630 RETURN
640 LET DT=5
650 IF EP<>P THEN LET B$="SEGUIR A QUIEN ?":RETURN
660 LET NP=FNA(20):GOSUB 950:LET P=NP
670 IF FNA(10)>8 THEN LET P=KP
680 IF FNA(10)>7 THEN LET B$="LE PERDISTE POR UN MOMENTO !":RETURN
690 LET EP=P
700 LET B$="LE TIENES A LA VISTA":RETURN
710 PRINT:PRINT "CUANTOS MINUTOS":INPUT DT
720 RETURN
730 PRINT:PRINT "DONDE QUIERES ENCONTRARLE":INPUT S$
740 PRINT:PRINT "A QUE HORA (HH.MM)"
750 INPUT U
760 IF P=CP AND T1<U AND H<CH THEN LET F(2)=1
770 LET DT=5:RETURN
780 LET B$="NADA AQUI":LET DT=10
790 IF P=KP THEN LET B$="HAS ENCONTRADO UNA LLAVE":LET F(5)=1
800 RETURN
810 LET DT=0:LET B$="LA HORA ES "+STR$(H)+","+STR$(M):RETURN
820 LET DT=5
830 IF U=0 THEN GOTO 860
840 PRINT:PRINT "EL LUGAR DE ENCUENTRO ES"
850 PRINT S$;" A LAS "U
860 GOSUB 1300:RETURN

```

190-200: Comprueba si la misión ha terminado.

210-270: Coge una palabra del jugador y comprueba si coincide con alguna de las que hay en la memoria. V es el número de esa palabra.

280-290: Comprueba el valor de NE para detectar si el jugador ha sido visto por el enemigo.

300: Salta a la subrutina, dependiendo de la palabra introducida.

310: Incrementa la hora.

330: Comprueba si el jugador ha consumido todo su tiempo.

340: Lleva al enemigo de regreso al hotel.

360-410: SUBRUTINA IR.

420-490: SUBRUTINA DECIR.

500-540: SUBRUTINA EXAMINAR.

550-570: SUBRUTINA LEER.

580-640: SUBRUTINA ABRIR.

650-700: SUBRUTINA SEGUIR.

710-720: SUBRUTINA ESPERAR.

730-770: SUBRUTINA DEJAR.

780-800: SUBRUTINA BUSCAR.

810: SUBRUTINA HORA.



```

870 LET DT=0:LET B$="PERDON?":RETURN
880 PRINT "DEMASIADO TARDE":STOP
890 PRINT:PRINT "BIEN HECHO. TU MISION HA SIDO UN EXITO"
900 LET TL=(FH-HI)*60-M
910 LET S=INT((20/MH+TL/120)*50)
920 PRINT:PRINT "TU CALIFICACION DE ESPIA"
930 PRINT "ES ";S
940 STOP
950 LET NY=INT((MP-1)/5)
960 LET NX=MP-5*NY
970 LET DX=ABS(X-MX):LET DY=ABS(Y-MY)
980 LET X=NX:LET Y=NY
990 LET D=SQR(DX*2+DY*2)
1000 LET DT=INT(5*D):RETURN
1010 DIM R$(20),V$(11),F(7)
1020 FOR I=1 TO 20:READ R$(I)
1030 NEXT I
1040 FOR I=1 TO 11:READ V$(I):NEXT I
1050 RETURN
1060 DATA "AEROPUERTO", "PARADA DE
AUTOBUS", "PUENTE", "CANAL", "IGLESIA"
1070 DATA "PARGUE", "CAFE", "BANCO", "CINE", "HOTEL"
1080 DATA "CASINO", "PLAZA MAYOR", "OFICINA
DE CORREOS", "COMISARIA", "FERIA"
1090 DATA "ESTACION", "AYUNTAMIENTO", "EMBAJADA",
"JARDINES", "CASTILLO"
1100 DATA "IR", "DECIR", "EXAMINAR", "LEER", "ABRIR", "SEGUIR", "ESPERAR", "DEJAR", "BUSCAR"
1110 DATA "HORA", "AYUDA"
1120 DATA "FLAN", "SALMON", "KOALA", "BOLLO", "CRUCIGRAMA", "CANGURO"
1130 LET H=FNA(2)+8:LET M=0
1140 LET FH=FNA(2)+14
1150 LET CH=FNA(2)+H
1160 CLS:PRINT:PRINT
1170 LET NE=0:LET T$="EL CONTACTO ESTA AGUI"
1180 LET MP=FNA(18):LET KP=FNA(18)
1190 LET EP=FNA(18):LET CP=FNA(18)
1200 FOR I=1 TO FNA(6)
1210 READ P$:NEXT I
1220 LET NL=FNA(1900)+99
1230 PRINT "CITA"
1240 PRINT "====":PRINT
1250 PRINT "RECOGER EL MENSAJE DE"
1260 PRINT R$(MP)
1270 PRINT "EL CONTACTO LO RECOGERA"
1280 PRINT "EN ";R$(CP)
1290 PRINT "A LAS ";CH;".00"
1300 PRINT "EL ULTIMO VUELO SALE"
1310 PRINT "A LAS ";FH;".00"
1320 PRINT:PRINT "PULSA RETURN PARA CONTINUAR"
1330 INPUT B$:RETURN

```



El símbolo ^  
significa «elevado a».  
En el VIC-20, C-64  
y Spectrum aparece  
como: ↑.

820-860: SUBROUTINA AYUDA.

870: SUBROUTINA «palabra no reconocida».

880: Mensaje «tiempo consumido».

890-940: Imprime el mensaje y el tanteo si la misión ha sido un éxito. El tanteo depende del número de movimientos realizados y del tiempo que sobre.

950-1000: Subrutina que calcula la distancia que el jugador ha recorrido y el tiempo utilizado en el movimiento.

1010-1050: Subrutina que lee los datos.

1060-1120: Líneas de datos.

1130: Elige la hora de comienzo.

1140: Elige la hora del último vuelo.

1150: Elige la hora en que el contacto recogerá el mensaje.

1180-1190: Elige las posiciones del mensaje, la llave, el enemigo y dónde recogerá el mensaje el contacto.

1200-1210: Elige la palabra clave.

1220: Elige el número de la caja de seguridad.

1230-1320: Imprime la introducción al juego.



Las líneas 1010-1320  
se llaman sección  
de «inicialización»  
del programa.

## Líneas de conversión

- 10 DEF FNA(X)=INT(RND\*X+1)
- 10 borrar esta línea
- 70,1160 Sustituir CLS por HOME
- ▲ 70,1160 Sustituir CLS por PRINT CHR\$(147)
- 235 IF LEN(I\$)<7 THEN LET I\$=I\$+" ":GOTO 235
- 300 80SUB 360\*(V=1)+420\*(V=2)+490\*(V=3)+540\*(V=4)+570\*(V=5)+640\*(V=6)+710\*(V=7)+730\*(V=8)+780\*(V=9)+810\*(V=10)+820\*(V=11)+870\*(V=12)
- 365 IF LEN(NG),14 THEN LET N\$=N\$+" ":GOTO 365
- 735 IF LEN(S\$),14 THEN LET S\$=S\$+" ":GOTO 735
- 1010 DIM R\$(20,14):DIM V\$(11,7):DIM F(7)
- 280,340,660,670,680,1130,1140,1150,1180,1190,1200,1220 Sustituir FNA( ) por RND( )  
Por ejemplo FNA(20) seria RND(20)



Procura averiguar qué hace exactamente cada una de las líneas de las subrutinas.

Las variables F(número) son «banderas». F(5), por ejemplo, es 1 si tienes la llave y 0, si no. Intenta encontrar las otras banderas y saber su función.



# Respuestas a las cuestiones planteadas

985

Aquí están las respuestas a algunas de las cuestiones planteadas a lo largo del libro. Tus respuestas pueden ser diferentes, pero

si funcionan no importa. Comprueba si son tan simples y funcionales como las que damos a continuación.

## Ojos de espía

Cómo añadir una rutina para conservar el mejor tanteo.

```
15 LET H=0
215 PRINT "MEJOR TANTEO = ";H
275 IF P>H THEN LET H=P:PRINT "ESTE ES EL MEJOR TANTEO"
290 INPUT A$:IF A$="S"
290 INPUT A$:IF A$="S" THEN GOTO 20
```

## El faro rastreador

Esta es la manera de hacer que la luz permanezca más tiempo encendida...

```
430 IF C1=TC THEN LET L=1:LET C1=0:LET
    TC=INT(RND(1)*10+(8-A))+10:GOSUB 300
```

...y ésta la de cómo dar al jugador una segunda oportunidad.

```
15 LET NG=0
255 LET NG=NG+1
257 IF NG<2 THEN FOR T=1 TO 200:NEXT T:GOTO 30
```



Aumenta este número si quieres más de dos partidas.



Este bucle produce una pausa para que puedas leer el mensaje. Si quieres que la pausa sea mayor, aumenta este número.



Usa la versión de RND que corresponda a tu ordenador.



NG es una nueva variable que contiene el número de partidas jugadas.

## El robot-espía

Estas líneas harán que el ordenador pregunte al jugador su nombre cuando alcance la mejor puntuación.

```
25 LET H$="NADIE"
330 Quita esta línea
365 IF E=1 AND S>H THEN LET H=S:GOSUB 410
410 PRINT "ESTE ES EL MEJOR TANTEO!"
420 PRINT "TECLEA TU NOMBRE"
430 INPUT H$:RETURN
```

## Examen de espía

Con las líneas siguientes el número de posiciones es 15 en vez de 10.

```
10 DIM N(15):DIM N$(5)
70,160,200,460 Cambiar el 10 de estas líneas por 15
```

# JUEGOS DE ORDENADOR

Cada uno de estos nuevos libros contiene sencillos programas de juegos para utilizar en los ordenadores domésticos de mayor difusión. Junto a cada programa hay explicaciones sobre cómo funciona, así como problemas y sugerencias para introducir cambios que compliquen las aventuras o las hagan más sencillas, según los deseos del usuario. Por medio de estos apasionantes juegos, incluso los que nunca hayan utilizado un ordenador aprenderán con facilidad cómo funciona un programa sencillo, y podrán diseñar alguno por sí mismos. Al final de cada libro aparecen las claves de conversión para diversos ordenadores, así como un glosario de los términos más usuales en BASIC.

## Títulos de la serie

